
 Appolonian sphere packing -- Derivation of "special" coordinates and inversion formula
 Author: Ronny Peikert

with(linalg):
 Abbreviations for square roots

q2 := 2^(1/2):  q3 := 3^(1/2):  q6 := 6^(1/2):
 Simplification procedure

mysimp := proc(e)
    expand(radsimp(simplify(e), ratdenom));
end:

 Inversion spheres, expressed in Cartesian coordinates plus radius, (x,y,z,r)

inv1c := [0, 0, 0, q3 - q2];
inv2c := [q3, -q3, -q3, 2*q2]; 
inv3c := [-q3,  q3, -q3, 2*q2];
inv4c := [-q3, -q3, q3, 2*q2]; 
inv5c := [q3, q3, q3, 2*q2]; 

inv1c := 0, 0, 0, 3 2

inv2c := 3 , 3 , 3 , 2 2

inv3c := 3 , 3 , 3 , 2 2

inv4c := 3 , 3 , 3 , 2 2

inv5c := 3 , 3 , 3 , 2 2

 Initial spheres (generators), expressed in Cartesian coordinates plus radius, (x,y,z,r)

init1c := [0, 0, 0, -1];
init2c := [q2-q3, q3-q2, q3-q2, q6-2 ];
init3c := [q3-q2, q2-q3, q3-q2, q6-2 ];
init4c := [q3-q2, q3-q2, q2-q3, q6-2 ];
init5c := [q2-q3, q2-q3, q2-q3, q6-2 ];

init1c := 0, 0, 0, 1

init2c := 3 2 , 3 2 , 3 2 , 2 6

init3c := 3 2 , 3 2 , 3 2 , 2 6

init4c := 3 2 , 3 2 , 3 2 , 2 6

init5c := 3 2 , 3 2 , 3 2 , 2 6

 Transformation from (x,y,z,r) to inversive coordinates (a,b,c,d,e) -- taken from Boyd's (?) papers

cartToInv := proc(xyzr)
    local x, y, z, r, a, b, c, d, e;
    x := xyzr[1]; y := xyzr[2]; z := xyzr[3]; r := xyzr[4];
    a := x/r;
    b := y/r;
    c := z/r;

    d := (x^2 + y^2 + z^2 - r^2 - 1) / (2*r);
    e := (x^2 + y^2 + z^2 - r^2 + 1) / (2*r);
    map(mysimp, [a, b, c, d, e]);
end:

 Back transformation

invToCart := proc(abcde)
    local a, b, c, d, e, x, y, z, r;
    a := abcde[1]; 
    b := abcde[2]; 
    c := abcde[3]; 
    d := abcde[4];
    e := abcde[5];
    r := 1 / (e - d);
    x := a * r;
    y := b * r;
    z := c * r;
    map(mysimp, [x, y, z, r]);
end:

 Transformation of initial spheres

init1i := cartToInv(init1c);
init2i := cartToInv(init2c);
init3i := cartToInv(init3c);
init4i := cartToInv(init4c);
init5i := cartToInv(init5c); 

Transformation from inversive coordinates (a,b,c,d,e) to "special" coordinates (A,B,C,D,E).  The matrix
T is chosen such that it maps the 5 basis vectors to the vectors init1i, ..., init5i

T := transpose(matrix([init1i, init2i, init3i, init4i, init5i]))
;

TI := inverse(T):

invToSpec := proc(abcde)
    map(mysimp, evalm(TI &* abcde));
end:

 Back transformation

specToInv := proc(ABCDE)
    map(mysimp, evalm(T &* ABCDE));
end:

 Compound transformation from Cartesian to special coordinates

cartToSpec := proc(xyzr)
    invToSpec(cartToInv(xyzr));
end:

 Back transformation

specToCart := proc(ABCDE)
    invToCart(specToInv(ABCDE));
end:

 Inversion of a sphere p at a sphere q, in inversive coordinates.

iproduct := proc(u, v)
    u[1]*v[1] + 
    u[2]*v[2] +
    u[3]*v[3] +
    u[4]*v[4] - 
    u[5]*v[5];
end:

reflectAt := proc(p, q)
    local r:
    r := p - 2*iproduct(p,q)*q;
end:

 Inversion of the sphere p at the sphere q, in special coordinates

reflectAt_s := proc(p, q)

    local pi, qi, ri, r;
    pi := specToInv(p);
    qi := specToInv(q); 
    ri := reflectAt(pi,qi);
    r := invToSpec(ri);
    map(mysimp, r);
end:

 Inversion spheres, in special coordinates

inv1s := cartToSpec(inv1c);
inv2s := cartToSpec(inv2c); 
inv3s := cartToSpec(inv3c); 
inv4s := cartToSpec(inv4c);
inv5s := cartToSpec(inv5c); 

 Initial spheres (generators), in special coordinates (= basis vectors by construction)

init1s := cartToSpec(init1c);
init2s := cartToSpec(init2c);
init3s := cartToSpec(init3c);
init4s := cartToSpec(init4c);
init5s := cartToSpec(init5c);

init1s := 1 0 0 0 0

init2s := 0 1 0 0 0

init3s := 0 0 1 0 0

init4s := 0 0 0 1 0

init5s := 0 0 0 0 1

 Spherical inversion (reflection at the 5 inversion spheres) of a general sphere, in special coordinates

  reflectAt_s([A,B,C,D,E], inv1s);
reflectAt_s([A,B,C,D,E], inv2s);
reflectAt_s([A,B,C,D,E], inv3s);
reflectAt_s([A,B,C,D,E], inv4s);
reflectAt_s([A,B,C,D,E], inv5s);

A B A C A D A E A

B A B B C B D B E

C A B C C C D C E

D A B D C D D D E

E A B E C E D E E

 Testing -- First generation of spheres (= initial sphere #i reflected at inversion circle #i), in special 
coordinates

sphere11s := reflectAt_s(init1s, inv1s);
sphere22s := reflectAt_s(init2s, inv2s);
sphere33s := reflectAt_s(init3s, inv3s);
sphere44s := reflectAt_s(init4s, inv4s);
sphere55s := reflectAt_s(init5s, inv5s);

sphere11s := 1 1 1 1 1

sphere22s := 1 1 1 1 1

sphere33s := 1 1 1 1 1

sphere44s := 1 1 1 1 1

sphere55s := 1 1 1 1 1

 ... which is in Cartesian coordinates:

evalf(specToCart(sphere11s));
evalf(specToCart(sphere22s));
evalf(specToCart(sphere33s));
evalf(specToCart(sphere44s));
evalf(specToCart(sphere55s));

0., 0., 0., 0.101020514

0.4099776108, 0.4099776108, 0.4099776108, 0.2898979486

0.4099776108, 0.4099776108, 0.4099776108, 0.2898979486

0.4099776108, 0.4099776108, 0.4099776108, 0.2898979486

0.4099776108, 0.4099776108, 0.4099776108, 0.2898979486

 Second generation of spheres, in special coordinates

  sphere112s := reflectAt_s(sphere11s, inv2s);
sphere113s := reflectAt_s(sphere11s, inv3s);
sphere114s := reflectAt_s(sphere11s, inv4s);
sphere115s := reflectAt_s(sphere11s, inv5s);
sphere221s := reflectAt_s(sphere22s, inv1s);
sphere223s := reflectAt_s(sphere22s, inv3s);

sphere112s := 0 1 2 2 2

sphere113s := 0 2 1 2 2

sphere114s := 0 2 2 1 2

sphere115s := 0 2 2 2 1

sphere221s := 1 0 2 2 2

sphere223s := 2 0 1 2 2

 etc.


